Obtaining Harmonics from Opera-2D Results

Animesh Jain
Brookhaven National Laboratory
Upton, New York 11973-5000, USA

US Particle Accelerator School on *Superconducting Accelerator Magnets*
Phoenix, Arizona, January 16-20, 2006

2-D Fields: Harmonic Series

$$B_r(r,\theta) = \sum_{n=1}^{\infty} \left(\frac{r}{R_{ref}} \right)^{n-1} [B_n \sin(n\theta) + A_n \cos(n\theta)]$$

$$B_\theta(r,\theta) = \sum_{n=1}^{\infty} \left(\frac{r}{R_{ref}} \right)^{n-1} [B_n \cos(n\theta) - A_n \sin(n\theta)]$$

$$B_y(r,\theta) = \sum_{n=1}^{\infty} \left(\frac{r}{R_{ref}} \right)^{n-1} [B_n \cos\{(n-1)\theta\} - A_n \sin\{(n-1)\theta\}]$$

$$B_x(r,\theta) = \sum_{n=1}^{\infty} \left(\frac{r}{R_{ref}} \right)^{n-1} [B_n \sin\{(n-1)\theta\} + A_n \cos\{(n-1)\theta\}]$$

$$\boxed{A_z(r,\theta) = \mathrm{Re}\left[-\int \boldsymbol{B}(\boldsymbol{z})\, d\boldsymbol{z}\right] = \sum_{n=1}^{\infty} \left(\frac{r}{R_{ref}} \right)^{n-1} \left(\frac{r}{n} \right) [A_n \sin(n\theta) - B_n \cos(n\theta)]}$$

Obtaining Harmonics from Field

The harmonic coefficients, B_n and A_n, can be obtained by Fourier analyzing ANY component of the field, OR by Fourier analyzing the vector potential, at a fixed radius, as a function of angle.

Exception: B_y and B_x are insensitive to A_1 and B_1 (dipole terms) respectively.

Vector Potential may be a good choice: Primary quantity obtained by Opera-2D. Fields are derived from the Vector Potential.

Harmonics from Vector Potential

$$A_z(r,\theta) = \sum_{n=1}^{\infty} \left(\frac{r}{R_{ref}} \right)^{n-1} \left(\frac{r}{n} \right) \left[A_n \sin(n\theta) - B_n \cos(n\theta) \right]$$

$$\int_0^{2\pi} \cos(n\phi)\cos(m\phi)\,d\phi = \pi\delta_{mn}\,; \int_0^{2\pi} \sin(n\phi)\cos(m\phi)\,d\phi = 0$$

$$\int_0^{2\pi} A_z(r,\theta)\cos(n\theta)(r\,d\theta) = -\pi B_n \left(\frac{r}{R_{ref}} \right)^{n-1} \left(\frac{r^2}{n} \right)$$

$$\int_0^{2\pi} A_z(r,\theta)\sin(n\theta)(r\,d\theta) = \pi A_n \left(\frac{r}{R_{ref}} \right)^{n-1} \left(\frac{r^2}{n} \right)$$

Command File for Opera-2D

abpot_gen.comi - Notepad

File Edit Format View Help

```
/ UNITS LENGTH=MM FLUX=TESLA FIEL=A/M DENS=A/MM2 ENERG=JOULE/MM
/ HARMONIC ANALYSIS FOR NO SYMMETRY IN THE MODEL.
/ ALL NORMAL AND SKEW TERMS ARE ALLOWED. INTEGRATION FROM 0 TO 2*PI
/ HARMONIC ANALYSIS BASED ON VECTOR POTENTIAL (UNIT=T.m ASSUMED)
/ #LUNI IS NO. OF LENGTH UNITS/METER (=100 FOR L IN CM.)
$CONS #LUNI 1000.
$CONS #RREF 25.0
$CONS #R   30.0
$CONS #PI 3.14159265358979323 84
$CONS #XOFF 0.0
$CONS #YOFF 0.0
$CONS #MAG 2
$PARA #N #MAG
$PARA #IKA SCALE
/NOTE: EXPRESSION FOR #GN DEPENDS ON LENGTH UNITS:
$PARA #GN  #N*(10000./#PI)*(#RREF/#R)**(#N-1)*#LUNI/(#R*#R)
$PARA #X X-#XOFF
$PARA #Y Y-#YOFF
/ Calculate the stored energy (integral of B.H):
INTA REG1=1,REG2=*
$CONS #STOR ENERGY1
/ Calculate Amplitude of fundamental term for normalization:
INTC RADI=#R, P1=0., P2=360., XCEN=#XOFF, YCEN=#YOFF,
COMP=POT*COS((#N)*ATAN2(#Y;#X)),ERRO=512
$CONS #BRFB -INTEGRAL/10000.
INTC COMP=POT*SIN((#N)*ATAN2(#Y;#X)),ERRO=512
$CONS #BRFA INTEGRAL/10000.
$CONS #BREF #GN*(#BRFA**2+#BRFB**2)**0.5
```

Command File for Opera-2D

abpot_gen.comi - Notepad

File Edit Format View Help

```
$prom name 'File Name (No Extension):'
$open 2 &name&.anbn
$format 1  CHAR 65
$format 2 CHAR 6
$format 4 stri 25 string=' n,        bn,         an'
$format 5 inte  2
$format 6 stri  2 string=', '
$FORMAT 7 FIXE  10 3
$FORMAT 8 FIXE  6 2
$FORMAT 9 EXPO 12
$assign 1
$write 2 'File=, &name&.anbn'
$assign 2 8 6 2 8 6 2 8 6 2 8
$write 2 'Xoff=,' #XOFF 'Yoff=,' #YOFF 'Rref=,' #RREF 'Rint=,' #R
$FORMAT 11 FIXE 10 5
$ASSIGN 2 11 6 2 11 6 2 11
$WRIT 2 'Bref=,' #BREF 'I_kA=,' #IKA 'Ener=,' #STOR
$assign 4
$write 2
$ASSIGN 5 6 7 6 7
/ Normal and skew harmonics from n=1 to 15:
$DO #N 1 15
INTC COMP=POT*COS((#N)*ATAN2(#Y;#X))
$CONS #BN -#GN*INTEGRAL/#BREF
INTC COMP=POT*SIN((#N)*ATAN2(#Y;#X))
$CONS #AN #GN*INTEGRAL/#BREF
$WRIT 2  #N  #BN  #AN
$END DO
$CLOSE 2
```